Buckling of Spring Supported Tapered Columns allowing for Shear Deformation

A. Watson[1] and W. P. Howson [2]

Summary

Theory and concise computer program are presented that enable elastic critical buckling loads of spring supported, shear sensitive, tapered columns to be determined accurately. The program is based on a stiffness model and necessitates the solution of a transcendental eigenvalue problem. It incorporates the Wittrick-Williams algorithm and thus ensures convergence to the lowest, or any other required buckling load. The program is fully described with illustrative examples.

Introduction

It has long been recognised that a substantial increase in buckling load can be achieved using a tapered column compared to its uniform counterpart of the same mass. Alternatively, less structure mass is required to sustain the equivalent load. This can be particularly important in the aerospace industry, where weight reduction is always an important consideration. Recent developments in materials technology have also lead to composite columns that are much more sensitive to shear deformation than their metallic counterparts. This is due to their low G/E ratio, which is typically 3 to 4 times less than a metallic column and can be as much as 10 times less [1]. The effect of shear deformation can now be very significant, even on the lowest critical buckling load.

The present paper reformulates existing buckling theory for a uniform member in a way that highlights the effect of the shear parameter. The governing differential equation is solved in terms of non-classical boundary conditions and the resulting equations are presented in stiffness matrix form. The stiffness matrix for the tapered member is then obtained by dividing the tapered member into a series of uniform members and assembling these into the required matrix. The computer program to implement this is efficient and is sufficiently concise that it can be readily understood. This ensures that it can be easily changed to accommodate individual needs. The program can handle a range of tapered or uniform single columns with any combination of boundary conditions in the form of spring supports. The effect of shear deformation can be allowed for or ignored and convergence to any required buckling load, to any required accuracy, is guaranteed by use of the Wittrick-Williams algorithm [2].

Theory

Figure 1 shows the forces and displacements associated with a typical elemental length, dx, of a uniform column that is subjected to a compressive axial force P. Resolving vertically and taking moments gives, respectively

$$-Q+\left(Q+\frac{dQ}{dx}dx\right)=0 \qquad Qdx+M-\left(M+\frac{dM}{dx}dx\right)+P\frac{dV}{dx}dx=0 \tag{1}$$

[1]Dept. Aero & Auto Engineering, Loughborough University, Loughborough, LE11 3TU, UK
[2]Cardiff School of Engineering, Cardiff University, Cardiff, CF24 0YF, UK



Figure 1: Forces acting on a typical elemental length of the member.



Figure 2: Spring supported, tapered member divided into 8 segments.

Simple bending theory and the shear relationships [2] give

$$M = -EI\frac{d\Psi}{dx} \qquad Q + P\frac{dV}{dx} = k'AG\Gamma \quad \text{and} \quad \Gamma = \frac{dV}{dx} - \Psi \tag{2}$$

where $I =$ second moment of area of the member cross-section, $A =$ cross-sectional area, $E =$ Young's modulus, $G =$ modulus of rigidity, $k' =$ section shape factor and Q, M, V and Ψ are the shear force due to bending, bending moment, lateral displacement, and bending slope, respectively, at a typical distance x from the left hand end of the member.

Eliminating either V, Ψ or Γ from equations (1)-(2) yields the differential equation governing the buckling of a uniform column subject to shear deformation as

$$\mathrm{D}^2[(1-s^2p^2)\mathrm{D}^2 + p^2]\Lambda = 0 \qquad (\mathrm{D} = d/d\xi) \tag{3}$$

where $\xi = x/L$, $L =$ member length, $s^2 = EI/k'AGL^2$, $p^2 = PL^2/EI$ and $\Lambda = V$, Ψ or Γ. This non-dimensional formulation is particularly convenient, since the effects of shear deformation are included if s^2 takes its natural value and are omitted when s^2 is set to zero. Equation (3) is a linear differential equation with constant coefficient whose solution can be found in standard form, subject to the following boundary conditions, see Figure 2,

$$\begin{array}{ll} M_1 \mid_{\xi=0} = -k_{\theta 1}\Psi_1 \mid_{\xi=0} & Q_1 \mid_{\xi=0} = -k_{\delta 1}V_1 \mid_{\xi=0} \\ M_2 \mid_{\xi=1} = -k_{\theta 2}\Psi_2 \mid_{\xi=1} & Q_2 \mid_{\xi=1} = -k_{\delta 2}V_2 \mid_{\xi=1} \end{array} \tag{4}$$

This leads to a stiffness relationship that may be stated as

$$\begin{bmatrix} Q_1L \\ M_1 \\ Q_2L \\ M_2 \end{bmatrix} = \frac{EI}{L}\begin{bmatrix} A_1 + K_{\delta 1} & A_2 & A_3 & A_4 \\ & A_5 + K_{\theta 1} & A_6 & A_7 \\ Symm & & A_8 + K_{\delta 2} & A_9 \\ & & & A_{10} + K_{\theta 2} \end{bmatrix}\begin{bmatrix} V_1/L \\ \Psi_1 \\ V_2/L \\ \Psi_2 \end{bmatrix} \tag{5}$$



Figure 3: Sample of cross sections covered. ϕ, ζ and η are constant along the length

where the stiffness coefficients A_1 - A_{10} are defined by equations (6) and (7) [2].

$$\begin{array}{ll} A_5 = A_{10} = \gamma(S - \beta C) & A_7 = \gamma(\beta - S) \\ A_3 = -A_1 = -A_8 = -\gamma\beta^2 S & A_6 = A_9 = -A_2 = -A_4 = -\gamma\beta(1 - C) \end{array} \tag{6}$$

$$\gamma = \frac{\alpha}{2(1-C) - \beta S} \quad \alpha^2 = \frac{p^2}{1 - s^2 p^2} \quad \beta^2 = p^2(1 - s^2 p^2) \quad C = \cos\alpha \quad S = \sin\alpha \tag{7}$$

and $K_{\delta 1}$, $K_{\theta 1}$, $K_{\delta 2}$, $K_{\theta 2}$ are non-dimensional support stiffnesses given by

$$K_{\delta 1} = \frac{k_{\delta 1} L^3}{EI} \quad K_{\theta 1} = \frac{k_{\theta 1} L}{EI} \quad K_{\delta 2} = \frac{k_{\delta 2} L^3}{EI} \quad K_{\theta 2} = \frac{k_{\theta 2} L}{EI} \tag{8}$$

The linear taper ($T_r > -1$) used in the following program can deal with the cross sections shown in Figure 3 in which $A(x)$ and $I(x)$ are given by

$$A(x) = A_0 \left(1 + T_r \frac{x}{L}\right)^2 \quad (A_0 = A(0)) \qquad I(x) = I_0 \left(1 + T_r \frac{x}{L}\right)^4 \quad (I_0 = I(0)) \tag{9}$$

Note that symmetry can sometimes be used to analyse doubly tapered members.

FORTRAN 77 **Computer Program**

The annotation to the right hand side of the code is merely to assist with understanding.

```
      DIMENSION A(10)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      PI=4.0*ATAN(1.0)                          ! π
      READ(5,*)NP                               ! No. of problems
      WRITE(6,1000)NP
      IP=0                                      ! Set problem number
10    IP=IP+1                                   ! Loop on problems
      WRITE(6,1010)IP
      READ(5,*)AI0,AA0,AL,P,E,G,SF,TR,CV        ! See Table 1
      WRITE(6,1020)AI0,AA0,AL,P,E,G,SF,TR,CV
      READ(5,*)JR,NS,SI,AKD1,AKT1,AKD2,AKT2
      WRITE(6,1000)JR,NS,SI,AKD1,AKT1,AKD2,AKT2
```

```
      BLFU=1.E10                          ! Set buckling load
      BLFL=0.0                            ! factor bounds
      BLF=1.0                             ! Set BLF
      EI0=E*AI0                           ! EI at LH end
      PE=PI*PI*EI0/(AL*AL)                ! Euler load, Pe
      SL=AL/FLOAT(NS)                     ! Segment length
20    PC=BLF*P                            ! Current trial load
      IG=2
      JB=0                                ! Set number roots passed
      A(1)=AKD1                           ! Set boundary conditions
      A(2)=0.0                            ! at LH end
      A(5)=AKT1
      X=-SL/2.0
      DO 70 IS=1,NS                       ! Loop on no. of segments
      X=X+SL                              ! Locate centroid
      FAC=(1.0+X*TR/AL)**2                ! Taper factor
      AA=AA0*FAC                          ! Centroidal area
      EI=EI0*FAC*FAC                      ! Centroidal I
      EIDL2=EI/(SL*SL)                    ! EI/L for segment
      P2=PC/EIDL2                         ! p^2 equation (3)
      S2=SI*EIDL2/(SF*AA*G)               ! s^2 equation (3)
      APA=SQRT(P2/(1.0-S2*P2))            ! α
      BTA=P2/APA                          ! β
      S=SIN(APA)                          ! sin α
      C=COS(APA)                          ! cos β
      BS=BTA*S
      GA=APA/(2.0*(1.0-C)-BS)             ! γ
      GB=GA*BTA
      GT=(GA*S-GB*C)*EIDL2*SL
      A(3)=-GB*BS*EIDL2/SL                ! Coefficients of A
      A(1)=A(1)-A(3)                      ! Assemble segments
      A(8)=-A(3)
      A(5)=A(5)+GT
      A(10)=GT
      A(6)=-GB*(1.0-C)*EIDL2
      A(7)=GA*(BTA-S)*EIDL2*SL
      A(2)=A(2)-A(6)
      A(4)=-A(6)
      A(9)=A(6)
      JB=JB+INT(APA/PI)                   ! Accumulate no. of
      IF(GT.LT.0.0)JB=JB-1                ! roots passed [2]
      IF((GT-A(7)*A(7)/GT).LT.0.0)JB=JB-1
      IF(IS.LT.NS)GOTO 30
      IG=3
      A(8)=A(8)+AKD2                      ! Set RH boundary
      A(10)=A(10)+AKT2                    ! conditions
30    DO 50 I=1,IG                        ! Start Gauss
      IPT=10-(4-I)*(7-I)/2                ! elimination
```

```
      IF(A(IPT).GT.1.0E28)GOTO 50            ! Branch on suppressed
      PT=1.0/A(IPT)                          ! freedom
      DO 40 J=I+1,4
      IPT=IPT+1
      PIV=A(IPT)*PT
      L=IPT-1
      J1=10-(4-J)*(7-J)/2
      J2=J1+4-J
      DO 40 K=J1,J2
      L=L+1
40    A(K)=A(K)-PIV*A(L)
      IF(IS.EQ.NS)IG=4
50    CONTINUE
      DO 60 I=1,IG
      II=10-(4-I)*(7-I)/2
60    IF(A(II).LT.0.0)JB=JB+1
      A(1)=A(8)                              ! Move A(8), A(9) and
      A(2)=A(9)                              ! A(10) elements to A(1),
70    A(5)=A(10)                             ! A(2) and A(5) locations
      IF(CV*(BLF-BLFL).LE.BLF)GOTO 100       ! End if converged
      IF(JB.LT.JR) GOTO 80                   ! Branch on looker bound
      BLFU=BLF                               ! Set upper bound
      GOTO 90
80    BLFL=BLF                               ! Set lower bound
      IF(BLFU.LT.1.E9)GOTO 90                ! Set new load factor
      BLF=2.0*BLF                            ! if no new upper bound
      GOTO 20
90    BLF=0.5*(BLFL+BLFU)                    ! Set new load factor
      GOTO 20                                ! Branch to next cycle
      STERM=0.0                              ! 1/s = 0 when shear
      IF(SI.GT.0.5.AND.TR.EQ.0)STERM=1.0/SQRT(S2)  ! not considered
100   WRITE(6,1030)BLF,PC,PC/PE,SQRT(PE/PC),STERM
      IF(IP.LT.NP)GOTO 10                    ! Loop on problems
      STOP
1000  FORMAT(1X,I4,1P5E9.2)
1010  FORMAT(/1X,"PROBLEM No.",I3)
1020  FORMAT(1X,1P8E9.2)
1030  FORMAT(1X,1P5E14.7)
      END
```

Data preparation and interpretation of results

The data input for the program is straightforward and is presented in Table 1. The output from the program consists of an echo of the input data followed by a single line of results, as described in Table 2. In order to consolidate the input/output scheme, an example of a data file is given in Table 3, while the corresponding output file is given in Table 4. The spring stiffness value $1.e30$ is recognised to be a clamping stiffness. There are two problems to be solved. The basic problem is the same in each case, except that the first one does not allow for shear deformation while the second one does.

Table 1: Data input scheme

Line	Variable	Comment		
1	NP	Number of problems.		
2	AI0	Second moment of area of cross-section, I, at LH end of member.		
	AA0	Area of cross-section, A, at LH end of member.		
	AL	Member length, L.	P	Initial axial load, P.
	E	Young's modulus, E.	G	Shear modulus, G.
	SF	Section shape factor, k'.	TR	Taper ratio ($0 =$ uniform member).
	CV	Solution accuracy 1 part in CV.		
3	JR	Number of buckling load required. 1 = lowest.		
	NS	Number of uniform segments by which the tapered member is divided.		
	SI	1.0 if shear considered, 0.0 otherwise.		
	AKD1	Lateral spring stiffness at LH end of member.		
	AKT1	Rotational spring stiffness at LH end of member.		
	AKD2	Lateral spring stiffness at RH end of member.		
	AKT2	Rotational spring stiffness at RH end of member.		

Table 2: Output results.

Item	Comment
BLF	Buckling load factor. PC = BLF*P, where P is the original axial load.
PC	Buckling load.
PC/PE	Buckling load / Euler load when $T_r = 0$.
$\sqrt{PE/PC}$	Effective length coefficient.
STERM	$1/s$, where s is defined below equation (3). Only relevant if $T_r = 0$

Table 3: Example of input data file.

```
2
8.e-4 2.e-2 5. 1.7e8 2.e11 8.e10 .7 .414 1.e6
1 512 0. 1.e30 1.e30 1.e9 0.
8.e-4 2.e-2 5. 1.7e8 2.e11 8.e10 .7 .414 1.e6
1 512 1. 1.e30 1.e30 1.e9 0.
```

Table 4: Output from data of Table 3.

```
PROBLEM No.1
1.5140142E+0 2.5738242E8 4.0747331E0
4.9539363E-1 0.0000000E0
PROBLEM No. 2
1.2802725E+0 2.1764632E8 3.4456536E0
5.3872136E-1 0.0000000E0
```

References

1. Bank, L.C. and Kao, C.H. (1989): "The influence of geometric and material design variables on the free vibration of thin-walled composite material beams", *Trans. ASME, J. of Vibration, Acoustics, Stress and Reliability*, Vol. 111, pp. 290-297.
2. Howson, W.P., Banerjee, J.R. and Williams, F.W. (1983):"Concise equations and program for exact eigensolution of plane frames including member shear", *Advances in Engineering Software*, Vol. 5, pp. 137-141.